DFUNIT – Getting started
Getting DFUnit
You can get the latest version (in fact, all versions) from https://github.com/olaeld/DFUnit  
[image: ]Download the ZIP or use your git client to clone the repository.
In order to run the example workspace, also download the zip / use git client to clone the DataFlex Patterns workspace https://github.com/olaeld/DataFlexPatterns and save this workspace on your system inside the DFUnit main directory, but as a sibling of the DFUnit workspace directory. 
Running DFUnit
From the “DFUnit Example workspace” folder, open the workspace DFUnitExampleWorkspace.sws with Studio 18.2.
Compile and run the project “DFUnit TestRunner”, and you’re good to go!
If all goes well, the program should start, the test fixtures should be executed, and the result box should become green. You can run the tests again by pressing the “Run tests” button.
So I ran the program. Now what?
Now it’s your turn to write tests. Let me walk you through some DFUnit essentials.
DFUnit essentials
Test fixtures
Tests belong in a test fixture. Each test fixture should test a set of related functionality. Ideally, one test fixture should test the functionality of one class.

Tests
A test fixture contains a set of tests.
For each public method of your class, you would normally write at least one test.
Assertions
A test should either fail or pass. Normally, this is done by assertion. If an assert fails while running a test, this means that the test has failed (the test result box will turn red).
Assert methods
Procedure Assert Boolean bCondition String sAssertMessage
Procedure AssertFalse Boolean bCondition String sAssertMessage
Procedure AssertIAreEqual Integer Expected Integer Actual String sAssertMessage
Procedure AssertNAreEqual Number Expected Number Actual String sAssertMessage
Procedure AssertSAreEqual String Expected String Actual String sAssertMessage
Procedure AssertDTAreEqual DateTime Expected DateTime Actual String sAssertMessage
An example test fixture
Filename: VDFPatterns\AppSrc\cArrayIteratorIntegerTests.pkg
Use DFUnit.pkg
Use cArrayIteratorInteger.pkg

Object cArrayIteratorIntegerTests is a cTestFixture
    Procedure Setup
        Object oIterator is a cArrayIteratorInteger
        End_Object
        
        Integer[] iArray
        Integer iCount
        For iCount from 0 to 19
            Move iCount to iArray[iCount]
        Loop
        Send AttachArray to oIterator iArray
    End_Procedure
    Procedure TearDown
        Send Destroy of oIterator
    End_Procedure
    Object IteratorReset_IteratorMoveNext_CurrentArrayValue_Returns0 is a cTest
        Procedure Test
            Send IteratorReset to oIterator
            Boolean bFoundFirstElement
            Get IteratorMoveNext of oIterator to bFoundFirstElement
            Send Assert bFoundFirstElement "First element of iterator not found"
            Send AssertIAreEqual 0 (CurrentArrayValue(oIterator(Self)))
        End_Procedure
    End_Object
    Object IteratorMoveNext_20times_ReturnsFalse is a cTest
        Procedure Test
            Send IteratorReset to oIterator
            Integer iCount
            Boolean bFound
            For iCount from 1 to 20
                Get IteratorMoveNext of oIterator to bFound
                Send Assert bFound ("Iterator should contain 20 elements, but found only" ;
                                    * (iCount - 1))
                If (not(bFound)) Procedure_Return
            Loop
            Get IteratorMoveNext of oIterator to bFound
            Send AssertFalse bFound "Iterator should not contain more than 20 elements"
        End_Procedure
    End_Object
End_Object
Running tests
[bookmark: _GoBack]To run the tests, include the test fixture file in oDFUnitTests.pkg.
DFUnit detects its presence, and the TestRunner program will execute its tests.
image1.png
gt
11 At of nitTesting Codle Bxamples
. DatsFlexpattems

s oFunit

3 DFURit Ecample Workspace:




